

# NAG Fortran Library Routine Document

## F07GNF (ZPPSV)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F07GNF (ZPPSV) computes the solution to a complex system of linear equations

$$AX = B,$$

where  $A$  is an  $n$  by  $n$  Hermitian positive-definite matrix stored in packed format and  $X$  and  $B$  are  $n$  by  $r$  matrices.

### 2 Specification

```
SUBROUTINE F07GNF (UPLO, N, NRHS, AP, B, LDB, INFO)
INTEGER          N, NRHS, LDB, INFO
complex*16      AP(*), B(LDB,*)
CHARACTER*1      UPLO
```

The routine may be called by its LAPACK name ***zppsv***.

### 3 Description

The Cholesky decomposition is used to factor  $A$  as  $A = U^H U$ , if  $UPLO = 'U'$  or  $A = LL^H$ , if  $UPLO = 'L'$ , where  $U$  is an upper triangular matrix and  $L$  is a lower triangular matrix. The factored form of  $A$  is then used to solve the system of equations  $AX = B$ .

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

- 1: UPLO – CHARACTER\*1 *Input*  
*On entry:* if  $UPLO = 'U'$ , the upper triangle of  $A$  is stored.  
 If  $UPLO = 'L'$ , the lower triangle of  $A$  is stored.  
*Constraint:*  $UPLO = 'U'$  or  $'L'$ .
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of linear equations, i.e., the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: NRHS – INTEGER *Input*  
*On entry:*  $r$ , the number of right-hand sides, i.e., the number of columns of the matrix  $B$ .  
*Constraint:*  $NRHS \geq 0$ .

- 4: AP(\*) – **complex\*16** array Input/Output

**Note:** the dimension of the array AP must be at least  $\max(N \times (N + 1)/2)$ .

*On entry:* the upper or lower triangle of the Hermitian matrix  $A$ , packed columnwise in a linear array. The  $j$ th column of  $A$  is stored in the array AP as follows:

if UPLO = 'U',  $AP(i + (j - 1) \times j/2) = a_{ij}$  for  $1 \leq i \leq j$ ;  
 if UPLO = 'L',  $AP(i + (j - 1) \times (2n - j)/2) = a_{ij}$  for  $j \leq i \leq n$ .

*On exit:* if INFO = 0, the factor  $U$  or  $L$  from the Cholesky factorization  $A = U^H U$  or  $A = LL^H$ , in the same storage format as  $A$ .

- 5: B(LDB,\*) – **complex\*16** array Input/Output

**Note:** the second dimension of the array B must be at least  $\max(1, \text{NRHS})$ . To solve the equations  $Ax = b$ , where  $b$  is a single right-hand side, B may be supplied as a one-dimensional array with length  $\text{LDB} = \max(1, N)$ .

*On entry:* the  $n$  by  $r$  right-hand side matrix  $B$ .

*On exit:* if INFO = 0, the  $n$  by  $r$  solution matrix  $X$ .

- 6: LDB – INTEGER Input

*On entry:* the first dimension of the array B as declared in the (sub)program from which F07GNF (ZPPSV) is called.

*Constraint:*  $\text{LDB} \geq \max(1, N)$ .

- 7: INFO – INTEGER Output

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ , the leading minor of order  $i$  of  $A$  is not positive-definite, so the factorization could not be completed, and the solution has not been computed.

## 7 Accuracy

The computed solution for a single right-hand side,  $\hat{x}$ , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and  $\epsilon$  is the **machine precision**. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where  $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$ , the condition number of  $A$  with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

F07GPF (ZPPSVX) is a comprehensive LAPACK driver that returns forward and backward error bounds and an estimate of the condition number. Alternatively, F04CEF solves  $Ax = b$  and returns a forward error bound and condition estimate. F04CEF calls F07GNF (ZPPSV) to solve the equations.

## 8 Further Comments

The total number of floating point operations is approximately  $\frac{4}{3}n^3 + 8n^2r$ , where  $r$  is the number of right-hand sides.

The real analogue of this routine is F07GAF (DPPSV).

## 9 Example

To solve the equations

$$Ax = b,$$

where  $A$  is the symmetric positive-definite matrix

$$A = \begin{pmatrix} 3.23 & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 \end{pmatrix}$$

and

$$b = \begin{pmatrix} 3.93 - 6.14i \\ 6.17 + 9.42i \\ -7.17 - 21.83i \\ 1.99 - 14.38i \end{pmatrix}.$$

Details of the Cholesky factorization of  $A$  are also output.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07GNF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
      CHARACTER        UPLO
      PARAMETER        (UPLO='U')
*      .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N
*      .. Local Arrays ..
      COMPLEX *16      AP((NMAX*(NMAX+1))/2), B(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
      EXTERNAL         X04DDF, ZPPSV
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F07GNF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*          Read the upper or lower triangular part of the matrix A from
*          data file
*
          IF (UPLO.EQ.'U') THEN
```

```

      READ (NIN,*) ((AP(I+(J*(J-1))/2),J=I,N),I=1,N)
    ELSE IF (UPLO.EQ.'L') THEN
      READ (NIN,*) ((AP(I+((2*N-J)*(J-1))/2),J=1,I),I=1,N)
    END IF
*
*   Read b from data file
*
    READ (NIN,*) (B(I),I=1,N)
*
*   Solve the equations Ax = b for x
*
    CALL ZPPSV(UPLO,N,1,AP,B,N,INFO)
*
    IF (INFO.EQ.0) THEN
*
*       Print solution
*
      WRITE (NOUT,*) 'Solution'
      WRITE (NOUT,99999) (B(I),I=1,N)
*
*       Print details of factorization
*
      WRITE (NOUT,*)
      IFAIL = 0
      CALL X04DDF(UPLO,'Non-unit diagonal',N,AP,'Bracketed',
+                'F7.4','Cholesky factor','Integer',RLABS,
+                'Integer',CLABS,80,0,IFAIL)
*
    ELSE
      WRITE (NOUT,99998) 'The leading minor of order ', INFO,
+        ' is not positive definite'
    END IF
    ELSE
      WRITE (NOUT,*) 'NMAX too small'
    END IF
    STOP
*
99999 FORMAT ((3X,4(' (',F7.4,',',F7.4,')',:)))
99998 FORMAT (1X,A,I3,A)
    END

```

## 9.2 Program Data

F07GNF Example Program Data

```

      4                                     :Value of N
    ( 3.23,  0.00) ( 1.51, -1.92) ( 1.90,  0.84) ( 0.42,  2.50)
              ( 3.58,  0.00) (-0.23,  1.11) (-1.18,  1.37)
                      ( 4.09,  0.00) ( 2.33, -0.14)
                                ( 4.29,  0.00) :End of matrix A
    ( 3.93, -6.14) ( 6.17,  9.42) (-7.17,-21.83) ( 1.99,-14.38) :End of vector b

```

## 9.3 Program Results

F07GNF Example Program Results

Solution

```

    ( 1.0000,-1.0000) ( 0.0000, 3.0000) (-4.0000,-5.0000) ( 2.0000, 1.0000)

```

Cholesky factor

```

      1           2           3           4
1  ( 1.7972, 0.0000) ( 0.8402,-1.0683) ( 1.0572, 0.4674) ( 0.2337, 1.3910)
2              ( 1.3164, 0.0000) (-0.4702,-0.3131) ( 0.0834,-0.0368)
3                      ( 1.5604, 0.0000) ( 0.9360,-0.9900)
4                          ( 0.6603, 0.0000)

```